X-ray Spectra Part I: Loading, Visualizing, & Grouping Data

Tiziana Di Salvo, <u>disalvo@fisica.unipa.it</u> Second Multiwavelength School, July 2010

This exercise presumes you have XSPEC and FTOOLS installed in your computer and you have got the data files cygx1_hard.tar.gz. Unzip and extract the data files from the tar file. Description of the datafiles:

cygx1_hard

lecs_60XY.pha	LECS spectrum (use it in the $0.1 - 4$ keV energy range)
lecs_bkg_8.pha	LECS background spectrum
lecs_sep97.rmf	LECS response matix
lecs_130_128_8_sep9	7.arf LECS effective area
mecs23_30XY.pha	MECS spectrum (use it in the 1.8-10 keV range)
mecs23_bkg_4.pha	MECS background spectrum
mecs1_sep97.rmf	MECS response matix
mecs23_4_sep97.arf	MECS effective area
hp_tot.pha	HPGSPC bkg subtracted spectrum (use it in the 7-30 keV range)
mtrhp_S.rmf	HPGSPC response matix
mtrhp_S.rmf	HPGSPC effective area
sys5_pds_256.pha pds_256_aug97.rmf	PDS bkg subtracted spectrum (use it in the 20-200 keV range) PDS response matix and effective area

Look at the content of your spectrum file with **fdump** (displays the file header and content or copies it in an output file) or **fv** (graphic interactive gui) or **grppha** (is an interactive command driven task to define or redefine and/or display and/or operate on fits files). Use **fhelp** or **fv** to obtain information on these F-tools.

Step 1 - Use grppha to look at responses, backgrounds, exposures, & backscal Use the following string command to look at your file spectrum, input.pha grppha input.pha output.pha

output.pha is the name of the output file spectrum if any modification is applied to the input file spectrum. All the information on responses, background, exposure and backscale are automatically displayed in your screen.

For instance it is possibile to add the response and ancillary files & background files to the header of the spectrum file in one of the following ways:

- Edit file with fv and change keyword RESPFILE or ANCRFILE or BACKFILE
- Open file with grppha and change keyword RESPFILE with the command: chkey RESPFILE filename.rmf
 Do the same for ANCREU E and RACKEU E

Do the same for ANCRFILE and BACKFILE

Use ex to exit grppha saving the information or quit if you do not want to save these modifications.

Step 2 - Apply a grouping or add a systematic error to the data using grppha

Use the following string command to look at your file spectrum, input.pha

grppha input.pha output.pha

Use the command show to display the current settings to the terminal. For instance use the command **show group** to display the current grouping or **show systematics** to display the current systematic error added to the data (if any).

Other examples: if you want to rebin your spectrum in order to have at least 20 counts per energy channel, use the command:

group min 20

or

group min 20 Nmin Nmax

where Nmin - Nmax are the minimum and maximum channel number to which you want to apply the grouping.

If you want to add a systematic error, for instance 1%, to all the data or to the channels from Nmin to Nmax, use the command:

systematic 0.01

or

systematic 0.01 Nmin Nmax

Do not forget to exit grppha with ex if you want to save these changes in the output.pha file spectrum.

Step 3 – Access xspec help

Edit the Xspec.init file in the .xspec directory in your home directory. If you want the help in a pdf file, you should go to the section "options and commands for displaying helpfiles" and use the following settings:

USE_ONLINE_HELP: false

LOCAL_HELP_FORMAT: pdf

Then give the appropriate PDF_COMMAND for your system (**acroread** or **xpdf** for Linux, **open** for Mac).

Step 4 – Open xspec and load the data

Open xspec in the directory with the cygx1_hard data. Load the LECS spectrum in the following way:

XSPEC12>data 1:1 lecs_60XY.pha	% Read the spectrum file
XSPEC12>back 1 lecs_bkg_8.pha	% Read the background spectrum
XSPEC12>resp 1 lecs_sep97.rmf	% Read the response matrix
XSPEC12>arf 1 lecs_130_128_8_sep97.arf	% Read the effective area file

With the command **show** you can have information on the loaded files (**show file**), response file (**show response**), source count rate (**show rate**), and so on. If you use the command

XSPEC12> show ?

You get information on all the parameters xspec is able to show. With the command

XSPEC12> show all

you get information on all the parameters of your data, fitting settings, and results.

Save the LECS data reading in order to facilitate the loading in future uses:

XSPEC12>save files lecs

This command will create a file lecs.xcm containing all the information for the data reading given above. You can look at the .xcm file editing it or using the command more:

XSPEC12>more lecs.xcm

If now you quit xspec and open it again, you can fastly read the lecs data with the command:

XSPEC12>@lecs.xcm

Step 5 – Plotting data

Open the plot window using

XSPEC12>cpd /xw

Plot counts versus channel energy:

XSPEC12>pl counts Plot counts versus energy:

XSPEC12>setplot energy XSPEC12>pl counts

This automatically sets the x-scale to a logarithmic scale. If you want to plot counts in logarithmic scale (both x and y) use

XSPEC12>pl lcounts

If you use **plot data** (or plot ldata) you can look at the rates versus channels or energy in a linear (or logarithimic) scale. Use the command **plot ?** to see all you can plot and the command

XSPEC12>help plot

to look at the help for the plot function. Use the commands:

XSPEC12>setplot wave

to plot counts (or rates) in a linear or logarithmic scale versus wavelength,

XSPEC12>setplot background

to look at the contribution from the background,

XSPEC12>setp reb 3 10

to rebin your spectrum for plotting purposes ONLY, that is for visual inspection. In this case, the first number is the minimum significance, that is adjacent bins are combined until they have a significant detection at least as large as 3 sigma. The second number is the maximum number of bins that may be combined (see help setpl for more info on the setpl command).

Step 6 - Save a hard copy of one of these plots

To modify a plot (set different scales, x and y ranges, labels, etc.) you can type

XSPEC12>iplot

to enter the qdp environment, and use all the qdp commands. Once you are happy with your plot and you want to save a copy, you can save it in a qdp file (which can be accessed and modified with qdp) with the following command:

PLT> we filename

This will create two files, filename.qdp and filename.pco, which can be used in qdp. Otherwise, you can save a hardcopy:

PLT> h filename.ps/cps

where cps is to get a color ps file.

Step 7 – Load a second data set

Load the MECS dataset in the following way:

XSPEC12>data 2:2 mecs23_30XY.pha

If ancillary and background files are not loaded automatically, load them in the following way:

XSPEC12>back 2 mecs23_bkg_4.pha

XSPEC12>resp 2 mecs1_sep97.rmf XSPEC12>arf 2 mecs23_4_sep97.arf

Two data set are now used and can be fitted simultaneously.

Step 8 – Select the correct energy range for fitting and define a model

The LECS response matrix is best calibrated in the energy range 0.12 - 4 keV, while MECS response matrix is best calibrated in the range 1.8 - 10 keV. To exclude the rest from the fitting procedure use the command "ignore":

XSPEC12>ig 1:**-0.12 XSPEC12>ig 1:4.0-** XSPEC12>ig 2:**-1.8 XSPEC12>ig 2:10.0-**

(if you use integer numbers xspec will interpret them as channels instead of energies). The command to do the opposite (that is to include in the fit a range that was previously excluded) is **notice**. Save the data loaded in a xcm file:

XSPEC12>save file lecs_mecs

A file lecs_mecs.xcm is created and can be used to easily re-load the data in a future session with the command @lecs_mecs in xspec. Plot the spectrum again to see if the correct energy range is now used.

Use the following command to look at all the available models in xspec

XSPEC12>help model

(you can also use **mo**? to get the list of model names)

Define a simple power-law model plus interstellar absorption in the following way:

XSPEC12>mo wabs * po

Give "return" to accept default input values of the parameters, or change them; you can change the starting value for the parameter, the step in the fitting, lower and upper limits to the parameter to be used in the fitting. For the second data set, give return if you want to fix the parameters' values to the corresponding value for the first data set, otherwise you can change their values if you want them to be fitted separately.

XSPEC12>renorm	% Simple renormalization of the model
XSPEC12>fit	% To fit the spectrum with the given model

Look at the χ^2 in order to valuate the goodness of the fit, and look at the residuals with

respect to this best fit model in order to evaluate where are the most significant residuals and how we can adjust the model in order to better fit the data.

XSPEC12>plot residuals	% Plot Data - Model
XSPEC12>plot ratio	% Plot Data / Model
XSPEC12>plot delchi	% Plot (Data – Model) / sigma
XSPEC12>plot ldata delchi	% Plot both the data with the folded model and the residuals in
	% units of sigma with respect to the folded model.

Add a normalization constant to take into account intercalibration differences between the two instruments. Let it to be free for the LECS data (1st data set), and fix it to 1 for the MECS data (2nd data set):

XSPEC12>addcomp 1 cons

The number 1 indicates the position where the new component has to be adeed. This is a multiplicative model component, so be sure that it will multiply all the components of the model. Type 1 for the input parameter for the 1st data set, and type 1 -1 for the input parameter for the 2nd data set; giving a negative number to the step will freeze the corresponding parameter in the fitting. Otherwise you can fix a parameter value with the command:

XSPEC12>freeze 5

where 5 is the parameter number you want to fix (use **thaw** if/when you want to make the parameter free; in this case the step will become again positive). You can use the command "show all" to look at fitting defaults, plot settings, all the data sets loaded, ignored channels, exposures, rates, model parameters, etc.

Use **show model** to be sure that the model components are in the desidered order, and **show parameter** to be sure that all the parameters are correctly defined.

Fit the spectrum again and plot the residuals; I suggest to **plot ldata delchi**, since in this case we have the information on how much data differ from the model in units of the corresponding error (in this case all the error bars in the residuals' plot are of size 1).

Step 9 – Spectral fitting and other kinds of plots

Residuals at 6.4 keV are clearly evident; add a Gaussian line to fit them:

XSPEC12>addcomp 3 ga us

Change the starting value for the normalization parameter to 1e-3. This can be done directly when the 3^{rd} parameter is shown or afterwards with the command:

XSPEC12>newpar 5

Then fit the spectrum again:

XSPEC12>fit

To plot over the data all the model components beside the total model use the command:

XSPEC12>setpl add XSPEC12> pl ldata delchi

To rescale the y axis of the upper panel (where the data and folded model and components are plotted), use the command **setpl com**; this allows you to use in xspec all the qdp commands. For instance:

XSPEC12>setp com r y 1e-2 40 XSPEC12> pl

It is also possibile to plot the so-called "unfolded" spectrum, that is to plot the spectrum without the effects of the instrumental response. Note that this can be done only for plotting purposes, since the effects of the instrumental response cannot be eliminated in any way. In XSPEC, this can only be done in reference to a fitted model; what is plotted is indeed the model, and the difference between each data point and the model is used to place the data points over the model more or less in the right position, but with respect to the given model. You can verify that the appearence of the data points will be different for different models, so these kind of plots can be used just for plotting purposes, but never to prove or disprove something (see **help plot** for more details).

XSPEC12>pl ufspec	% Plot the unfolded spectrum, the total model, and the
	% contributions to the model of the various additive components.
XSPEC12>pl eufspec	% Plot the unfolded spectrum and the model as $Ef(E)$, or, if
	% wavelength plotting has been set, $\lambda f(\lambda)$.
XSPEC12>pl eeufspec	% Plot the unfolded spectrum and the model as $E^2f(E)$, or, if
	% wavelength plotting has been set, $\lambda^2 f(\lambda)$.

You can also plot the model itself with **plot model** or **emodel** or **eemodel** with same meaning as before. Finally, it can also be useful to plot the total instrumental response efficiency versus incident photon energy (only currently noticed channels are included in the calculation of the efficiency). This is useful since discrete features in the effective area are difficult to calibrate and can give rise to systematic features in the spectrum (especially for bright sources). Hence, if you find unidentified features in your spectrum, always check the efficiency plot to check if discrete (miscalibrated) features are present in the instrumental response.

If you are now satisfied with your results, you can save the best fit model:

XSPEC12>save model filename

or you can save both data files and the final model:

X-ray Spectra Part II: Spectral analysis

Step 10 – Error calculation

Open xspec and load the xcm file containing both the data sets and the best fit model. The errors on the best fit parameters can be calculated only once the fitting has reached a minimum in the χ^2 . Let us calculate the uncertainty on the interstellar absorption Nh, par 2:

XSPEC12>err 2

The default gives the uncertainties at 90% confidence level for a single parameter (corresponding to $\Delta \chi^2 = 2.706$. If we want to calculate the errors on all the free parameters at 1 sigma confidence level, use:

XSPEC12>err 1.0 1-8

all the frozen parameters are skipped for the error calculation. During the error calculation xspec performs an accurate scanning of the parameters' space and it is possible that it finds new minima of the χ^2 . In this case, it will stop the error calculation and will reset all the parameters to the new values. In this case, the error calculation has to be restarted and the new parameters' values have to be checked and saved.

Step 11 – Calculate fluxes, luminosities, equivalent width

To calculate the source flux on the basis of the best fit model, use the command:

XSPEC12>flux 1 10

in this case the flux will be calculated in the 1-10 keV range. Note that fluxes may be different for different instruments due to the intercalibration constant factor. Use as reference the flux derived from the instrument for which the constant has been frozen to 1.

To calculate the luminosity you need to give the distance to the source. Assume a distance of 2 kpc to Cyg X-1. Translate this distance to a redshift, given by the relation: z = H0 * D / c, where H0 is the Hubble constant (default in xspec is H0 = 70 km/s/Mpc), D is the distance in Mpc and c the speed of light in km/s. This means that z = 70 * 0.002 / 300,000 = 4.7e-7. To obtain the source luminosity at the distance of 2 kpc and in the range 2-10 keV, use the command:

XSPEC12> lumin 2 10 4.7e-7

The energy range redshifted to the observed range must be contained by the range covered by the current spectra (which determine the range over which the model is evaluated). To extrapolate the model luminosity over a wider range, for instance 0.1 - 100 keV, define a response in the given range, and then re-calculate the luminosity:

XSPEC12>dummy 0.1 100 XSPEC12>lumin 0.1 100 4.7e-7

Calculate the equivalent width of an emission line (Gaussian or diskline or rdblur*gauss):

XSPEC12>eqw 3

where 3 is the model component number.

Step 12 – Change a model component

Suppose you want to fit the iron line with a diskline instead of a Gaussian. Remove the Gaussian component from you best fit model:

XSPEC12>del 3

where 3 indicates the model component number. Then add a diskline component:

XSPEC12>addcomp 3 diskline

(see help model diskline for a help on this model and a description of the input parameters). Fit again the spectrum, calculate the errors on the parameters and compare the results with respect to what is obtained with a Gaussian line.

If we consider the diskline model for the iron line as a complex development of the Gaussian model, we can compare the results with an F-test. It can be obtained in xspec in the following way

XSPEC12>ftest chi2 dof2 chi1 dof1 (where chi2 < chi1)

Step 12 – How to use a convolution model

Instead of diskline it is possible to use a Gaussian line convolved with relativistic effects, that is for instance convolved with the same kernel used for diskline that is called rdblur. In this way we obtain the same result as fitting with diskline. If you want to try this, remove the diskline component, add a Gaussian line with sigma fixed at 0, and convolve it with rdblur:

XSPEC12>del 3 XSPEC12>addc 3 gauss (give 0 -1 to the Gaussian sigma parameter, give 1e-3 to the Gaussian normalization) XSPEC12>addc 3 rdblur Fit the spectrum again and note that the results are the same obtained with diskline.

Step 13 – Generate statistic surfaces for 1 or more parameters

Sometimes the χ^2 space is so complicated, with several secondary minima, that the errors on the parameters are difficult to calculate or a greater control is needed than given with the error command. In this case, the command **steppar** has to be used. For example, for our last best fit model, lets check the behavior of the inner disk radius as derived from the iron line. This can be done with the command (use **help steppar** to get the help):

XSPEC>steppar 4 6.0 21.0 15

This will step par #4 from 6.0 Rg to 21.0 Rg in 15 steps (that is with a step of 1.0 Rg).

To see how the error bars of two or more parameters are correlated, we can run steppar for these two parameters and plot the corresponding contours. For instance, for the diskline model we can run steppar for the inner radius and the inclination in the following way:

XSPEC12>steppar 4 6.0 50.0 15 6 10 80 20

This will step par#4 from 6 to 50 Rg in 15 steps and par#6 from 10 to 80 in 20 steps. Use the command

XSPEC12>plot contour

to see the contour plot at three confidence levels for these two parameters. The three contours correspond to Delta fit statistic of 2.3, 4.61, 9.21, respectively.

XSPEC12>plot contour,,3,1.0,2.706,9.21

In this case three levels will be plotted corresponding to a Delta fit statistic of 1.0, 2.706, 9.21 (corresponding to 1 sigma, 90%, and 3 sigma confidence level, respectively). For more details see **help plot**.

NOTES:

BACKSCAL keyword

When XSPEC subtracts the background spectrum from the source spectrum, XSPEC normalizes the background spectrum in the following way:

Background_subtracted_spectrum (cts/s) = SOURCE_COUNTS/SOURCE_EXPOSURE -BGD_COUNTS/BGD_EXPOSURE/BGD_BACKSCAL*SOURCE_BACKSCAL.

Here, SOURCE_BACKSCAL and BGD_BACKSCAL are the BACKSCAL keyword values in the source and background spectral files respectively, and SOURCE_EXPOSURE and BGD_EXPOSURE are the EXPOSURE keyword values.

In Xselect/Extractor, when a spectral file is created, the BACKSCAL value is defined as:

(THE DETECTOR AREA FROM WHICH THE SPECTRUM IS EXTRACTED)/(TOTAL DETECTOR AREA).

Note that it is assumed that the background is proportional to the detector area. This is not precisely true, but is considered a simple and effective approximation.

The Scientific Instrumentation: BeppoSAX - Occhialini X-ray Astronomy Satellite

The X-ray Astronomy Satellite, named BeppoSAX to honor Professor Giuseppe ``Beppo" Occhialini, was launched on April 30th 1996. One of the main characteristics of this mission is the unique capability to observe X-ray sources over more than three decades of energy, from 0.1 keV to 200 keV, with good energy resolution in the whole range, temporal resolution of 16 µs, and imaging capabilities in the energy range 0.1-10 keV (Boella et al. 1997a). This is obtained using a system of four co-aligned instruments, called Narrow Field Instruments (NFI).

These consist of:

- A Low Energy Concentrator Spectrometer, LECS. It is a thin window position sensitive proportional counter with extended low energy response in the band 0.1-10 keV (Parmar et al. 1997). Its thin entrance window of 1.25 micron multilayer polyimide allows to extend the energy range down to 0.1 keV. However, due to UV contamination problems, the LECS is operated only at satellite night time, resulting in a reduced exposure time with respect to the other instruments. The energy resolution of the LECS is approximately given by: $\Delta E / E = 8.0 (E / 6 \text{ keV})^{-1/2} \%$ FWHM, where ΔE is the FWHM of the energy dispersion at the energy E. The LECS effective area is about 50 cm² at 7 keV.
- Two (originally three) **Medium Energy Concentrator Spectrometers, MECS.** They are position sensitive proportional counters operating in the 1.3-10 keV band (Boella et al. 1997b). The energy resolution of the MECS is the same of the LECS, and the effective area is about 100 cm² at 7 keV (two units).
- A **High Pressure Gas Scintillation Proportional Counter (HPGSPC**; energy range of 7--60 keV; Manzo et al. 1997). It is a new development of a gas scintillation proportional

counter, filled with a mixture of xenon and helium at the pressure of 5 atmosphere. This allows the detector to be sensitive between 3 and 120 keV. The energy resolution is approximated by: $\Delta E / E = 11.0 (E / 6 \text{ keV})^{-1/2} \%$ FWHM and the geometrical collecting area is 450 cm².

A Phoswich Detection System (PDS; energy range of 13-200 keV; Frontera et al. 1997). It consists of four independent NaI(Tl)/CsI(Na) phoswich scintillation detectors. Each of the four detectors is made of two crystals of NaI(Tl) and CsI(Na) forming what is known as PHOSWICH (acronym of PHOsphor sandWICH). The scintillation light produced in each phoswich is viewed by a photomultiplier tube. The energy resolution is of PDS is:
Δ E / E = 15.0 (E / 60 keV)^{-1/2} % FWHM and the effective area is of 500 cm² at 60 keV.

X-ray images were obtained with LECS and MECS, that are in the focal plane of an X-ray telescope, which in turn consists of a set of 30 nested conical mirror shells providing an approximation to a Wolter I type geometry. The LECS and MECS data were extracted from circular regions centered on the source, with radii of 8' and 4', corresponding to 90 % of the instruments' point spread function. Data extracted from the same detector regions during blank field observations were used for background subtraction. The HPGSPC and PDS are non-imaging instruments, with fields of view (about 1 degree FWHM) limited by rocking collimators that can alternate on and off the source. Several methods are used to discriminate background events providing a very low background level in both the HPGSPC and PDS.

Background subtraction was obtained in the PDS from data accumulated during the off-source intervals. The same technique was used for the background subtraction of the HPGSPC data in the analysis of Cyg X--1. In more recent observations the background subtraction for the HPGSPC was obtained by using spectra accumulated from Dark Earth data, instead of from off-source data (in this way the HPGSPC because the collimator was kept on source for the whole observation (doubling the on-source exposure time).

In the spectral analysis we have to use data corresponding to the following energy ranges: 0.12-4 keV for the LECS, 1.8-10 keV for the MECS, 7-40 keV for the HPGSPC and 15-200 keV for the PDS, since in these energy ranges the response matrix is best calibrated. A systematic error of 1% was added to each spectrum. As customary, in the spectral fitting procedure we have to use different normalizations for the LECS, HPGSPC and PDS spectra relative to the MECS spectra, and checking *a posteriori* that derived values were in the standard range for each instrument (see the BeppoSAX handbook http://www.sdc.asi.it/software).

Spectral Analysis

The spectral analysis consists in a fitting of the energy spectrum emitted by a source with spectral models, in order to disentangle its spectral components. This procedure is complicated by the fact that the measured X-ray spectrum is different from the real photon spectrum emitted by the source, because it is the convolution of the source photon spectrum, Nph(E), with a function which characterizes the energy dependent efficiency of the detecting instrument, i.e. its ``response'' function. The response function is given by the product of the detector ``effective'' collecting area, A(E), and the probability R(E,E') that a photon of energy E is detected in the energy channel E'. The measured spectrum also contains an additional term, Bi, corresponding to the detector background. Thus, for each energy channel i (corresponding to the energy E'), the detected rate of photons, Ci, is given by:

$$C_{i} = B_{i} + \int_{E'_{i}}^{E'_{i} + \Delta E'_{i}} dE' \int_{0}^{\infty} R(E, E') A(E) N_{ph}(E) dE$$

where $\Delta E'i$ is the energy width of the channel. Since both **R** and **A** are complex matrices, the equation above cannot be directly solved to find the source spectrum. The usual approach is to assume a model for the source spectrum, to use the equation to convolve this model with the detector response, and to compare the result with the measured spectrum. Using a fitting procedure one can minimize the difference between the model and the data, and find the best fit to the data. The goodness of the fit is quantified with the χ^2 -test, where:

$\chi^2 = \Sigma_i \left[(\text{Di-Mi})^2 / \sigma i^2 \right]$

where Di = Ci - Bi is the background subtracted measured spectrum, Mi is the model spectrum (which depends on a number of parameters), and σi^2 is the variance of the data in the channel i, containing both statistical and systematic uncertainties. It can be shown that $\chi^2 / n \sim 1$ for a fit with a high confidence, where n = N - p is the number of degrees of freedom (N is the total number of energy channels and p the number of estimated model parameters). The observed spectra have to be rebinned (i.e. adjacent channels added together) in order to have at least 20 counts in each energy channel (otherwise the χ^2 statistics cannot be used). The energy spectra are also rebinned in order to have approximately the same number of bins per instrument resolution element across the entire energy range.

<u>F-test</u>

Sometimes a ``more complex" model, obtained from a ``simple" model by the addition of **i** parameters, gives a better fit. To test if the improvement of the fit with respect to the simple model is real or a chance, one can use the F-test (see e.g. Bevington & Robinson 1992, see also Protassov et al. 2002, ApJ, 571, 545).

It consists in calculating the F-function:

 $F(n,i) = \{ [\chi^2(n) - \chi^2(n+i)]/i \} / \{ \chi^2(n+i)/(n+i) \}$

where **n** are the degrees of freedom. The higher is the obtained value of F(n,i), the smaller is the probability of chance improvement of the fit. Hence, this probability can be used to quantify the improvement of the fit when a further component is introduced in the spectral model.